Polynomes et matrices (exercice)

On considère les matrices réelles

$$A=\begin{pmatrix}0&0&1\\1&0&1\\0&1&-1\end{pmatrix} \quad \text{et} \quad I=\begin{pmatrix}1&0&0\\0&1&0\\0&0&1\end{pmatrix}$$

Pour tout polynôme réel $P=a_nX^n+..+a_1X+a_0$ on pose $P(A)=a_nA^n+...+a_1A+a_0I$.

1/ Soit $P=X^3+X^2-X-1$.

a) Factoriser P dans $\mathbf{R}[X]$.

b) Vérifier que $P(A)=0$, en déduire deux diviseurs de zéro dans l'anneau $\mathscr{M}_3(\mathbf{R})$ des matrices 3×3 à coefficients réels.

2/ On pose

$$Q=\frac{1}{4}(X^2+2X+1) \text{ et } R=-\frac{1}{4}(X^2+2X-3) \text{ et } S=-\frac{1}{4}(X^3+3X^2-X-3)$$

$$B=Q(A), \quad C=R(A), \quad D=S(A)$$

a) Calculer les matrices B, C et D.

b) Calculer $B-C+D$.

c) Calculer les produits B^2, C^2, D^2, BC, CB, BD, DB, CD, DC.

Les calculs qui suivent répondent aux questions posées ou viennent appuyer les démonstrations.

```
> load("polymat.mc")$
```

```
polymat(P,A) := block (
    [n,S,i,PP],
    PP:expand(P),
    n:hipow(PP,X),
    S:coeff(PP,X,n)*A^^n,
    for i:n-1 step -1 thru 0 do S:S+coeff(PP,X,i)*A^^i,
    S
);
```

```
> A:matrix([0,0,1],[1,0,1],[0,1,-1]);
```

$$\begin{pmatrix} 0 & 0 & 1 \\ 1 & 0 & 1 \\ 0 & 1 & -1 \end{pmatrix}$$

```
> P:X^3+X^2-X-1$
> P1:factor(P);
```

$$(X-1)\,(X+1)^2$$

```
> polymat(P,A);
```

$$\begin{pmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix}$$

```
> block(A1:polymat(part(P1,1),A),A2:polymat(part(P1,2),A),[A1,A2]);
```

$$\left[\begin{pmatrix} -1 & 0 & 1 \\ 1 & -1 & 1 \\ 0 & 1 & -2 \end{pmatrix}, \begin{pmatrix} 1 & 1 & 1 \\ 2 & 2 & 2 \\ 1 & 1 & 1 \end{pmatrix}\right]$$

```
> A1 . A2;
```

$$\begin{pmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix}$$

```
> Q:1/4*(X^2+2*X+1)$
> B:polymat(Q,A);
```

$$\begin{pmatrix} \frac{1}{4} & \frac{1}{4} & \frac{1}{4} \\ \frac{1}{2} & \frac{1}{2} & \frac{1}{2} \\ \frac{1}{4} & \frac{1}{4} & \frac{1}{4} \end{pmatrix}$$

```
> R:-1/4*(X^2+2*X-3)$
> C:polymat(R,A);
```

$$\begin{pmatrix} \frac{3}{4} & -\frac{1}{4} & -\frac{1}{4} \\ -\frac{1}{2} & \frac{1}{2} & -\frac{1}{2} \\ -\frac{1}{4} & -\frac{1}{4} & \frac{3}{4} \end{pmatrix}$$

```
> S:-1/4*(X^3+3*X^2-X-3)$
> D:polymat(S,A);
```

$$\begin{pmatrix} \frac{1}{2} & -\frac{1}{2} & \frac{1}{2} \\ 0 & 0 & 0 \\ -\frac{1}{2} & \frac{1}{2} & -\frac{1}{2} \end{pmatrix}$$

```
> B-C+D;
```

$$\begin{pmatrix} 0 & 0 & 1 \\ 1 & 0 & 1 \\ 0 & 1 & -1 \end{pmatrix}$$

```
> B^^2;
```

$$\begin{pmatrix} \frac{1}{4} & \frac{1}{4} & \frac{1}{4} \\ \frac{1}{2} & \frac{1}{2} & \frac{1}{2} \\ \frac{1}{4} & \frac{1}{4} & \frac{1}{4} \end{pmatrix}$$

```
> C^^2;
```

$$\begin{pmatrix} \frac{3}{4} & -\frac{1}{4} & -\frac{1}{4} \\ -\frac{1}{2} & \frac{1}{2} & -\frac{1}{2} \\ -\frac{1}{4} & -\frac{1}{4} & \frac{3}{4} \end{pmatrix}$$

```
> D^^2;
```

$$\begin{pmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix}$$

```
> B . C;
```

$$\begin{pmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix}$$

```
> C . D;
```

$$\begin{pmatrix} \frac{1}{2} & -\frac{1}{2} & \frac{1}{2} \\ 0 & 0 & 0 \\ -\frac{1}{2} & \frac{1}{2} & -\frac{1}{2} \end{pmatrix}$$

```
> D . B;
```

$$\begin{pmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix}$$